Pensieve Header: Semi-Symmetrized calculus in the 2D quotient.

Continues “A 2D B-Picture, IV.nb”, uses computations from “BCH in Blobs.nb”.

In[5]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\w-Computations"]
```

Out[5]=
```
C:\drorbn\AcademicPensieve\Projects\w-Computations
```

In[6]:=
```
ar[i_, j_] := t[i] h[j]
```

In[7]:=
```
HMultiply[x_, y_, z_][mix_] := Module[
  {ξ, η, j},
  ξ = D[mix, h[x]] /. t[s_] :> c[s];
  η = D[mix, h[y]] /. t[s_] :> c[s];
  j[ζ_] := If[ζ === 0, 1, (ⅇ^ζ - 1)/ζ];
  (mix /. {h[x] → 0, h[y] → 0}) +
   j[ξ] h[z] D[mix, h[x]]/j[ξ + η] + ⅇ^ξ j[η] h[z] D[mix, h[y]]/j[ξ + η]
 ]
```

In[8]:=
```
BDisplay[expr_] := Assuming[
  And @@ ((# > 0) & /@ Union[Cases[expr, _c, Infinity]]),
  Collect[expr, _h, Collect[#, _t, FullSimplify] &]
 ]
```

In[9]:=
```
HMultiply[3, 4, 5][t[1] h[3] + t[2] h[4]] // BDisplay
```

Out[9]=
$$h[5]\left(\frac{(-1+e^{c[1]})\,(c[1]+c[2])\,t[1]}{(-1+e^{c[1]+c[2]})\,c[1]}+\frac{e^{c[1]}\,(-1+e^{c[2]})\,(c[1]+c[2])\,t[2]}{(-1+e^{c[1]+c[2]})\,c[2]}\right)$$

In[10]:=
```
a = t[1] h[4] + t[2] h[5] + t[3] h[6]
```

Out[10]=
```
h[4] t[1] + h[5] t[2] + h[6] t[3]
```

In[11]:=
```
HMultiply[4, 5, 7][a] // BDisplay
```

Out[11]=
$$h[7]\left(\frac{(-1+e^{c[1]})\,(c[1]+c[2])\,t[1]}{(-1+e^{c[1]+c[2]})\,c[1]}+\frac{e^{c[1]}\,(-1+e^{c[2]})\,(c[1]+c[2])\,t[2]}{(-1+e^{c[1]+c[2]})\,c[2]}\right)+h[6]\,t[3]$$

In[12]:=
```
HMultiply[7, 6, 8][HMultiply[4, 5, 7][a]] // BDisplay
```

Out[12]=
$$h[8]\left(\frac{(-1+e^{c[1]})\,(c[1]+c[2]+c[3])\,t[1]}{(-1+e^{c[1]+c[2]+c[3]})\,c[1]}+\frac{e^{c[1]}\,(-1+e^{c[2]})\,(c[1]+c[2]+c[3])\,t[2]}{(-1+e^{c[1]+c[2]+c[3]})\,c[2]}+\frac{e^{c[1]+c[2]}\,(-1+e^{c[3]})\,(c[1]+c[2]+c[3])\,t[3]}{(-1+e^{c[1]+c[2]+c[3]})\,c[3]}\right)$$

In[13]:=
```
HMultiply[4, 7, 8][HMultiply[5, 6, 7][a]] // BDisplay
```

Out[13]=
$$h[8]\left(\frac{(-1+e^{c[1]})\,(c[1]+c[2]+c[3])\,t[1]}{(-1+e^{c[1]+c[2]+c[3]})\,c[1]}+\frac{e^{c[1]}\,(-1+e^{c[2]})\,(c[1]+c[2]+c[3])\,t[2]}{(-1+e^{c[1]+c[2]+c[3]})\,c[2]}+\frac{e^{c[1]+c[2]}\,(-1+e^{c[3]})\,(c[1]+c[2]+c[3])\,t[3]}{(-1+e^{c[1]+c[2]+c[3]})\,c[3]}\right)$$

In[14]:=
```
(HMultiply[7, 6, 8][HMultiply[4, 5, 7][a]] -
   HMultiply[4, 7, 8][HMultiply[5, 6, 7][a]]) // BDisplay
```

Out[14]= 0

In[15]:=
```
HFactorize[z_, xtails_List → x_, y_][mix_] := Module[
   {ytails},
   ytails = Complement[
     Union[Cases[mix, t[s_] ⧴ s, Infinity]],
     xtails
    ];
   HFactorize[z, x, ytails → y][mix]
  ];
HFactorize[z_, x_, ytails_List → y_][mix_] := Module[
  {ξ, ξ0, α, α0, β, β0},
  ξ0 = D[mix, h[z]];
  α0 = ξ0 /. ((t[#] → 0) & /@ ytails);
  β0 = ξ0 - α0;
  {ξ, α, β} = {ξ0, α0, β0} /. t[s_] ⧴ c[s];
  Expand[mix - ξ0 h[z] + Log[(α E^ξ + β)/ξ]/α α0 h[x] - Log[(α + β E^-ξ)/ξ]/β β0 h[y]]
 ]
```

In[17]:=
```
HMultiply[3, 4, 5][ar[1, 3] + ar[2, 4]] // BDisplay
```

Out[17]= $h[5]\left(\frac{(-1+e^{c[1]})\,(c[1]+c[2])\,t[1]}{(-1+e^{c[1]+c[2]})\,c[1]}+\frac{e^{c[1]}\,(-1+e^{c[2]})\,(c[1]+c[2])\,t[2]}{(-1+e^{c[1]+c[2]})\,c[2]}\right)$

In[18]:=
```
HMultiply[3, 4, 5][ar[1, 3] + ar[2, 4]] // HFactorize[5, {1} → 3, 4] //
  FunctionExpand // BDisplay
```

Out[18]= $h[3]\,t[1]+h[4]\,t[2]$